



fischertechnik®

COMPUTING
COMPUTING
COMPUTING
COMPUTING

Interface Commodore Computer



Inhalt

Einführung	3
Anschluß des Interface	4
fischertechnik computing Software	5
Diagnoseprogramm	8
Commodore BASIC	10
Checkliste	10
Technische Daten	10
Benutzung von fischertechnik Elektromechanik und Elektronik	11
Verdrahtungsplan	12

fischertechnik computing Interface

Lieber fischertechnik-Freund,

um mit einem Computer, in Erweiterung seiner Einsatzmöglichkeiten, auch technische Modelle ansteuern zu können, wurde fischertechnik computing entwickelt. Hierzu gehören sowohl der fischertechnik computing Baukasten und die fischertechnik computing Bausätze ebenso wie die fischertechnik computing Interfaces und die Software. Es ist jetzt möglich, technische Funktionen und Vorgänge zu simulieren, Aufgaben zu lösen und einfach viel Spaß an computergesteuerten Modellen zu haben.

Was braucht man zum Steuern der Modelle? Zunächst einmal das fischertechnik Modell zur Ausführung der Abläufe. Dann einen Heim- oder Personalcomputer, wie Sie ihn besitzen. Er dient der Steuerung und der Koordination. Und dann noch ein Interface als Bindeglied zwischen beiden.

Was Sie in den Händen halten, ist das fischertechnik computing Interface. Steuersignale, die von dem Computer kommen, z. B. „Motor einschalten!“, wer-


den von dem Interface in kräftige Ströme umgesetzt, die in der Lage sind, tatsächlich einen Motor zu bewegen. Wir sprechen in diesem Fall von einer Ausgabe. Die gedachte Blickrichtung verläuft von dem Computer nach außen. Aber auch der umgekehrte Weg ist denkbar und kommt vor. Die Modelle besitzen Taster, Potentiometer etc., um dem Computer Bericht zu erstatten, was an dem Modell draußen vorgeht. Auch hier greift das Interface wieder helfend ein und bereitet diese Signale dergestalt auf, daß sie eine für den Computer verständliche Eingabe darstellen.

Das fischertechnik Interface besitzt nun folgende Leistungsmerkmale:

- Mit ihm lassen sich vier fischertechnik Motoren, Lampen, Elektromagnete etc. steuern.
- Mit ihm kann man acht Taster oder Schalter abfragen.
- Darüber hinaus liefern zwei Eingänge die Werte von stufenlosen Signalgebern wie etwa Potentiometern.

Doch was würden alle elektrischen Verbindungen zwischen Computer und fischertechnik Modell mit Hilfe des Interface nutzen, wenn Sie keine Hilfsmittel hätten, jene zu aktivieren. Die Rede ist von der Software. Dieser Teil liegt in der Form einer Diskette vor. Auf ihr befindet sich ein Programm, das den Sprachschatz Ihres Computers derart erweitert, daß die Steuerung über das Interface tatsächlich erfolgen kann. Dieses Programm wird die Keimzelle Ihrer eigenen Programme sein. Doch damit nicht genug: Damit Sie den Einsatz dieser neuen Hilfsmittel studieren und lernen können, sind Beispielprogramme für alle fischertechnik computing Modelle auch noch untergebracht.

Sie sehen, es wartet eine ganze Menge von interessanten Aufgaben auf Sie. Ich wünsche Ihnen viel Spaß dabei. Ihr



Anschluß des Interface

Das fischertechnik Interface Commodore C64/VC20, Art.-Nr. 30562 und fischertechnik Interface Commodore 4xxx/8xxx, Art.-Nr. 30561, passen an nahezu alle Commodore Computer, denn sie werden an den Benutzer E/A Anschluß (user-port) angeschlossen. Dieser Anschluß ist an dem guten alten PET2001 ebenso vorhanden wie an dem CBM8296 und der Heimcomputerlinie VC20 und C64. Natürlich besitzt auch der SX64, die tragbare Variante des C64, den gleichen Anschluß. Der Unterschied besteht lediglich in der in dem nächsten Kapitel zu besprechenden Software. Hier können wir nur die neueren Produktlinien unterstützen. Besitzer von Commodore Computern aus der 2000er- und 3000er-Serie sind also aufgerufen, die entsprechenden Adaptionen vorzunehmen. Was die Zukunft betrifft, wird fischertechnik computing versuchen, ebenfalls die Software zu liefern, wenn der Computer die Benutzer E/A Schnittstelle besitzt. Der genannten Bedingung wegen müssen wir allerdings gleich die Computer C16 und 116 ausklammern.

Zum Anschluß des fischertechnik computing Interface gehen Sie folgendermaßen vor:

- Vergewissern Sie sich, daß der Computer abgeschaltet ist.
- Drehen Sie den Computer herum, so daß die rückseitige Anschlußleiste vor Ihnen liegt. Suchen Sie nun die Benutzer E/A Schnittstelle. Bei C64 und VC20 ist dies die ganz rechte Leiste, siehe Abbildung. Bei den System-Computern ist sie deutlich mit „user-port“ beschriftet.
- Entladen Sie eine eventuell vorhandene elektrostatische Aufladung durch Berühren der metallischen Teile des Gehäuse des Computers oder eines geerdeten Gegenstandes, z.B. einer Heizung.
- Legen Sie sich nun das Interface zurecht. An dem Interface ist ein Kabel mit einem Verbindungsstecker befestigt.

- Setzen Sie nun den Verbindungsstecker so an der vorhin georteten Verbindungsleiste an, daß die meisten Anschlüsse auf die Unterseite der Verbindungsleiste kommen. Der Stecker ist gegen Verpolung geschützt, da die Trennrippe zu dem Schlitz in der Verbindungsleiste passen muß.
- Kontrollieren Sie noch einmal, daß der Stecker seinen Weg finden wird, und drücken Sie nun den Stecker auf die Leiste.
- Schließen Sie nun das fischertechnik computing Interface an das fischertechnik Netzgerät mot4 an. Das Interface erwartet Gleichspannung zwischen 6 und 10 Volt. Verbinden Sie also eine der mit ⊕ gekennzeichneten Buchsen mit der seitlichen ⊕ Buchse des Netzgeräts, ebenso verfahren Sie mit der ⊖ Leitung. Welches Buchsenpaar des Interface Sie verwenden, ist gleichgültig. Die Anschlußbuchsen sind doppelt ausgeführt, da mit dem fischertechnik Netzgerät nie viel mehr als zwei Motoren gleichzeitig angesteuert werden sollen. Bei größeren Modellen muß daher mit zwei Netzgeräten eingespeist werden. Die Netzgeräte sind dann entkoppelt.

Die älteren fischertechnik Netzgeräte haben seitlich einen Wechselspannungsausgang. In diesem Fall müssen Sie den vorderen regelbaren Gleichspannungsausgang verwenden, den Sie aber bis zum Anschlag aufdrehen. Bei allen fischertechnik Netzgeräten liegt die ⊖ Buchse nun da, wo der Zeiger des Drehreglers hinzeigt. Eine Sorge können wir Ihnen gleich nehmen: Falls Sie einmal die Spannungsanschlüsse vertauschen, erleidet weder Ihr Interface noch Ihr Computer einen Schaden. Das Interface funktioniert halt nur nicht mit verpolter Versorgung.


- Verbinden Sie das fischertechnik computing Modell mit dem Interface. Hierzu dient das den Bausätzen und dem Baukasten fischertechnik computing beigefügte zwanzigadrige Flachband-

kabel. Dieses Kabel ist auch als Einzelteil aus dem Service-Set fischertechnik erhältlich.

- Die Reihenfolge, in der Sie das Interface und den Computer nun einschalten, spielt keine Rolle. Wenn Sie das Interface mal nicht benutzen und mit anderen Programmen arbeiten, brauchen Sie das Interface dennoch nicht abkabeln. Lassen Sie in diesem Fall einfach das Interface ausgeschaltet.
- Nebeneffekte des Interface: Solange das Interface eingesteckt ist, können Sie selbstverständlich keine Geräte benutzen, die auch an diesen Anschluß eingesteckt gehörten. Hierzu gehört das V24/RS232 Modul und das Telefonmodem.

Beim Arbeiten mit den Modellen sollten Sie sicherheitshalber immer zuerst eine eventuell vorhandene elektrostatische Aufladung ableiten, indem Sie einen geerdeten metallischen Gegenstand berühren, z.B. eine Heizung.

Doch nun genug von der Hardware, im nächsten Abschnitt wollen wir uns die fischertechnik computing Software vornehmen.



fischertechnik computing Software

Wer sich schon einmal mit dem Gedanken befaßt hat, irgendwelche Geräte oder Modelle mit dem Computer zu steuern, wird aus eigener Erfahrung wissen oder von anderen Computerfreunden gehört haben, daß dies alles gar nicht so einfach sei. Man brauche eine genaue Kenntnis des Computers, des Mikroprozessors und Ein-/Ausgabebausteins sowie der Maschinensprache für diese Aufgabe. Bislang hatten Sie damit recht und dadurch wurde leider auch mancher von diesem interessanten Kapitel der Computerei abgehalten. Dies muß nun nicht mehr der Fall sein. In dem Lieferumfang des Interface sind die vorliegende Anleitung und Programme auf Diskette enthalten. Davon ist ein Programm besonders wichtig, das Grundprogramm.

Wenn Sie das Interface noch nicht an Ihren Computer wie in dem vorigen Kapitel beschrieben angeschlossen haben, so sollten Sie es nun nachholen. Schalten Sie den Computer ein und legen Sie die fischertechnik computing Diskette in das Diskettenlaufwerk. Geben Sie ein:

```
LOAD " 0 : * " , 8  
RUN
```

In wenigen Sekunden erscheint eine Bildschirmmeldung und das Inhaltsverzeichnis der Diskette. Dabei werden nur die für Ihren Computer geeigneten Programme dargestellt. Der Ausdruck ist so gewählt, daß Sie nur noch mit dem Cursor hochfahren müssen und das gewünschte Programm mit der Return-taste anwählen können.

An dieser Stelle wollen wir in einem kurzen Einschub an jene Computerfreunde denken, die kein Diskettenlaufwerk oder eines mit einem anderen Aufzeichnungsformat besitzen. Wir müssen Sie leider bitten, sich mit Ihrer fischertechnik computing Diskette auf den Weg zu einem guten Freund zu machen. Wenn dieser eine Commodore-Anlage besitzt, die diese Bedingung erfüllt, so können Sie alle Programme

einzelnen einladen und z.B. auf die Datensette abspielen. Mit den Programmen auf Kassette können Sie nun zu Hause auch arbeiten, da die fischertechnik computing Programme selbst das Diskettenbetriebssystem nicht benötigen. Wenn alle Stricke reißen, so finden Sie die Programmlisten in dieser und in anderen fischertechnik computing Anleitungen zum Abtippen.

Laden Sie nun den Computer mit dem Grundprogramm GRUNDPR... Die Punkte stehen für eine Namensergänzung, die den Computertyp, für den sich das Programm eignet, angibt. Also z.B.:

```
LOAD " GRUNDPR . 64 " , 8
```

Wenn Sie nun das Kommando RUN eingeben, wird sich der Computer nach einer kurzen Gedankenpause wieder mit seinem Bereitzeichen melden. Äußerlich scheint sich nichts geändert zu haben. Dennoch besitzt Ihr Commodore nun einige neue Befehle, die vorher in dieser Form nicht in BASIC enthalten waren. Diese Befehle sind genau auf den Commodore Computer und das fischertechnik computing Interface abgestimmt. In dem Grundprogramm stecken nun die oben erwähnten Detailkenntnisse über die Computerbausteine in Maschinensprache. Statt detaillierten Computerkenntnissen brauchen Sie nun nur noch die folgend beschriebenen BASIC-Befehle zu beherrschen: Der Motorausgang M1 wird angesteuert mit:

```
SYS M1, EIN      SYS M1, AUS  
SYS M1, RECHTS  SYS M1, LINKS
```

Die Kommandoparameter bezeichnen den Motor und die Betriebsart. Die entsprechenden Befehle mit M2, M3 und M4 steuern die übrigen drei Ausgänge. Außerdem sollten Sie sich merken, daß EIN ebenfalls immer Rechtslauf bewirkt.

Die zehn Eingänge werden mit Hilfe der USR-Funktion des BASIC-Interpreters abgefragt.

USR(E1)

ist 1, wenn der Eingang E1 des Interface mit +5V verbunden ist. Sonst zeigt USR(E1) den Wert 0. Entsprechend erhält man mit USR(E2)... USR(E8) die Zustände der übrigen Digitaleingänge. Die Analogeingänge EX und EY werden über je ein Potentiometer (4,7 kΩ) mit +5V verbunden. Die Funktionen

USR(EX) USR(EY)

haben dann einen Wert zwischen 0 und 255, je nach Stellung der Potentiometer. Wird z.B. ein Roboterarm von einem Motor angetrieben und synchron mit der Bewegung des Arms das Potentiometer EX verstellt, so kann das Programm, indem es immer wieder die Funktion

USR(EX)

aufruft, die Bewegung des Roboters genau verfolgen. Achtung VC20-Besitzer: Benutzen Sie allerdings diese Kommandos nie, wenn an der betreffenden Leitung kein Potentiometer angeschlossen ist! Der Computer wartet sonst für immer auf das Signal und Sie können ihn daraus nur mit der Tastenkombination STOP und RESTORE erlösen. In der gleichen Situation wird sich bei den System-Computern ein Überlauf bei 255, bei dem C64 bei 65535 ergeben.

Der letzte der neuen Befehle ist

SYS INIT

Dieser wird benutzt, um das Interface in einen wohldefinierten Anfangszustand zu versetzen. Er kann auch benutzt werden, wenn alle Motorkanäle mit einem Male abgeschaltet werden sollen. Doch nun genug der langen Vorrede. Schließen sie einen fischertechnik Motor über das zwanzigpolige

farbcodierte Flachbandkabel an M1 an. Dies sind die gelbe und orange Leitung in der oberen Hälfte des Flachbandkabels. Geben Sie ein:

SYS M1, EIN

Der Motor wird kurz anlaufen und dann wieder stehenbleiben. Genießen Sie diesen Augenblick, er hat Ihnen das Gefühl gegeben, in kurzer Zeit die kompliziertesten fischertechnik Anlagen mit Ihrem Commodore Computer zu steuern.

Doch zunächst interessiert uns auch die Frage, wieso der Motor wieder stehenblieb. Hatten wir ihn nicht eingeschaltet? Gibt es zum Ausschalten nicht, wie oben beschrieben, einen eigenen Befehl? Nun, der Motor ist zwar stehengeblieben, aber in dem Interface ist nach wie vor gespeichert, daß er eigentlich laufen sollte. Das Interface hat sich selbst „schlafen gelegt“. Dies tut es immer, wenn innerhalb einer halben Sekunde kein neuer Befehl kommt. Es geschieht aus Sicherheitsgründen. Stellen Sie sich vor, Sie erproben ein neues Programm. Die Wahrscheinlichkeit, daß noch irgendwo ein Fehler im Programm versteckt ist, grenzt an Gewißheit. Der Computer bleibt mit einer leidigen Meldung wie

?SYNTAX ERROR IN...

stehen. Der Motor, der kurz vorher eingeschaltet wurde, bliebe jedoch nicht stehen und schicke sich an, das schöne Modell zu demolieren. Sie müßten zum Netzgerät hasten und schnell die Spannung abstellen.

Wie beruhigend ist es da, zu wissen, daß der Motor von alleine stehenbleiben wird. Auch dann, wenn Sie mit dem Tastendruck STOP den Programmablauf unterbrechen.

Und wenn es wieder weitergeht (z.B. mit CONT), so wird mit dem ersten Befehl das Interface wieder „aufgeweckt“ und hat keinen der Motoren vergessen. Der Ablauf kann weitergehen, als sei nichts geschehen.

Daß das Interface mit dem Abschalten nicht sofort zur Hand ist, wurde mit Bedacht gewählt. Zwischen den Ein- und Ausgabebefehlen an das Interface werden sich immer wieder Pausen aufgrund von Berechnungen ergeben, die es zu überbrücken gilt. Ob das Interface durch Ein- oder Ausgabebefehle aktiviert wird, können Sie auch durch einen Blick auf die Leuchtdiode des Interface sagen. Sie dient nicht nur der Spannungsanzeige, sondern auch der Betriebsanzeige.

Nun wollen wir noch einen kurzen Blick auf die Eingabebefehle werfen. Schließen Sie zwischen E1 (der braunen Leitung am unteren Rand) und +5V (der roten Leitung in der Mitte des Flachbandkabels) einen Taster an.

Probieren Sie aus:

PRINT USR(E1)

Je nachdem, ob der Taster zwischen E1 und +5V bei der Betätigung der Return-Taste des Computers gedrückt war oder nicht, wird auf dem Bildschirm eine 1 oder eine 0 ausgegeben. Wenn an dem Ausgang noch von vorhin der Motor angeschlossen war, wird er sich wieder rühren. Auch Eingabebefehle aktivieren wieder die Ausgänge des Interface. Nun schließen Sie bitte ein Potentiometer 4,7 k Ω zwischen EX und +5V an. Stellen Sie den Schleifer in eine mittlere Stellung und geben Sie ein

PRINT USR(EX)

Die Zahl, die jetzt auf dem Bildschirm erschienen ist, muß zwischen 0 und 255 liegen. Sofern Sie die Diskette nicht benutzen oder auf Kassette umkopieren konnten und das Grundprogramm von Hand eingegeben haben, sollten Sie es jetzt auf Kassette oder Diskette abspeichern. Sie werden es immer wieder brauchen, weil jedes Programm, das mit dem fischertechnik computing Interface Modelle steuern soll, mit diesem Vorspann beginnt, der die neuen Befehle installiert.

Damit wir das Potentiometer leichter beobachten können, wollen wir nun das erste fischertechnik computing Programm schreiben. Das Grundprogramm befindet sich in dem Computer und belegt die Zeilennummern 1 bis 500. Geben Sie daher ein:

```
510 PRINT USR(EX)  
520 GOTO 510  
RUN
```

Es dauert einen kurzen Moment, bis das Grundprogramm geladen ist und dann geht es los. Im Nu wird der Bildschirm mit Zahlen gefüllt, die ständig nach oben hinausgeschoben werden. Wenn Sie jetzt das Potentiometer in die Hand nehmen und den Schleifer drehen, werden Sie die Veränderung der Zahlen beobachten. Drehen Sie von einem Anschlag zum andern. Die eingelesenen Zahlen sollten zwischen 0 und 255 liegen. Die 0 und die 255 werden eventuell nicht ganz erreicht.

Zum Beenden des Programms müssen Sie STOP drücken.

Für diejenigen, die etwas genauer die Abläufe verstehen wollen und nicht nur die auf der Diskette vorliegenden Programme benutzen wollen, halten wir hier nun noch Detailinformationen bereit. Die Funktion des Grundprogramms besteht darin, in einen Speicherbereich des Commodore Computers ein kurzes Maschinenprogramm einzuschreiben. Dieses liegt codiert in den DATA-Zeilen vor. Beim Einschreiben wird gleich noch anhand einer Prüfsumme kontrolliert, ob sich bei der Übertragung der Zahlenwerte ein Fehler eingeschlichen hat. Das Maschinenprogramm belegt bei den verschiedenen Computertypen unterschiedliche Speicherbereiche.

Bei dem VC20 und den Systemcomputern 4xxx und 8xxx wird der Kassettenspeicher als Speicher benutzt. Damit entsteht kein Verlust des verfügbaren Speichers oder ein Konflikt mit anderen Systemprogrammen. Allerdings kann ein fischertechnik computing

```

10 REM INTERFACE PROGRAMM FUER COMMODORE 64
20 REM COPYRIGHT (C) ARTUR FISCHER FORSCHUNG 1984
30 REM AUFRUF DES PROGRAMMS MIT
40 REM SYS M1,EIN      SYS M1,AUS
50 REM SYS M1,RECHTS  SYS M1,LINKS
60 REM USR(E1)  USR(EX)  USR(EY)
70 REM M1 BIS M4 SIND MOTORANSTEUERUNGEN
80 REM E1 BIS E8 SIND DIGITALEINGAENGE
90 REM EX UND EY SIND ANALOGEINGAENGE
100 DATA 52736,169,0,240,38,169,3,208,10,53573
110 DATA 169,12,208,6,169,48,208,2,54395
120 DATA 169,192,120,133,255,32,253,174,55723
130 DATA 165,254,5,255,133,254,32,158,56979
140 DATA 183,138,37,255,133,255,165,254,58399
150 DATA 69,255,133,254,168,169,63,141,59651
160 DATA 3,221,162,8,169,48,6,254,60522
170 DATA 144,2,9,4,141,1,221,9,61053
180 DATA 8,141,1,221,202,208,237,169,62240
190 DATA 57,141,1,221,132,254,88,96,63230
200 DATA 120,32,161,183,224,162,240,51,64403
210 DATA 224,146,240,47,134,255,169,50,65668
220 DATA 141,1,221,9,8,141,1,221,66411
230 DATA 162,8,10,44,1,221,16,2,66875
240 DATA 9,1,160,48,140,1,221,160,67615
250 DATA 56,140,1,221,202,208,235,37,68715
260 DATA 255,168,240,2,160,1,32,162,69735
270 DATA 179,88,96,169,255,141,4,221,70888
280 DATA 141,5,221,169,185,141,14,221,71985
290 DATA 142,1,221,162,58,142,1,221,72933
300 DATA 173,4,221,162,3,202,208,253,74159
310 DATA 56,237,4,221,208,242,162,56,75345
320 DATA 142,1,221,56,169,255,237,4,76430
330 DATA 221,168,169,255,237,5,221,32,77738
340 DATA 145,179,76,137,206,0,0,0,78481
350 DATA 1,2,4,8,16,32,64,128,78736
360 DATA 162,146,255,170,85,85,80,206,79925
370 READ INIT : M1=INIT
380 FOR M3=0 TO 24 : FOR M2=0 TO 7
390 READ M4 : POKE INIT+M3*8+M2,M4
400 M1=M1+M4 : NEXT
410 READ M4 : IF M1<>M4 THEN PRINT"DATAFEHLER
IN ZEILE";M3*10+100:END
420 NEXT
430 READ E1,E2,E3,E4,E5,E6,E7,E8
440 M1=M1+E1+E2+E3+E4+E5+E6+E7+E8
450 READ M4 : IF M1<>M4 THEN PRINT"DATAFEHLER
IN ZEILE 350" : END
460 READ EX,EY,AUS,LINKS,RECHTS,EIN,M2,M3
470 M1=M1+EX+EY+AUS+LINKS+RECHTS+EIN+M2+M3
480 READ M4 : IF M1<>M4 THEN PRINT"DATAFEHLER
IN ZEILE 360" : END
485 POKE785,M2 : POKE786,M3
490 M1=INIT+4 : M2=M1+4 : M3=M2+4 : M4=M3+4
500 SYS INIT

```

```

10 REM INTERFACE PROGRAMM FUER VC20
20 REM COPYRIGHT (C) ARTUR FISCHER FORSCHUNG 1984
30 REM AUFRUF DES PROGRAMMS MIT
40 REM SYS M1,EIN      SYS M1,AUS
50 REM SYS M1,LINKS    SYS M1,RECHTS
60 REM USR(E1)  USR(EX)  USR(EY)
70 REM M1 BIS M4 SIND MOTORANSTEUERUNGEN
80 REM E1 BIS E8 SIND TASTEREINGAENGE
90 REM EX UND EY SIND POTENTIOMETEREINGAENGE
100 DATA 828,169,0,240,38,169,3,208,10,1665
110 DATA 169,12,208,6,169,48,208,2,2487
120 DATA 169,192,120,133,255,32,253,206,3847
130 DATA 165,254,5,255,133,254,32,158,5103
140 DATA 215,138,37,255,133,255,165,254,6555
150 DATA 69,255,133,254,168,169,63,141,7807
160 DATA 18,145,162,8,169,48,6,254,8617
170 DATA 144,2,9,4,141,1,221,9,9087
180 DATA 8,141,1,221,202,208,237,169,10213
190 DATA 57,141,1,221,132,254,88,96,11142
200 DATA 120,32,161,215,224,162,240,51,12345
210 DATA 224,144,240,47,134,255,169,50,13688
220 DATA 141,16,145,9,8,141,16,145,14229
230 DATA 162,8,10,44,16,145,16,2,14632
240 DATA 9,1,160,48,140,16,145,160,15311
250 DATA 56,140,16,145,202,208,235,37,16350
260 DATA 255,168,240,2,160,1,32,162,17370
270 DATA 211,88,96,169,32,141,27,145,18279
280 DATA 169,255,141,24,145,141,25,145,19324
290 DATA 142,16,145,162,56,142,16,145,20148
300 DATA 173,24,145,162,3,202,208,253,21318
310 DATA 56,237,24,145,208,242,56,169,22455
320 DATA 255,237,24,145,168,169,255,237,23945
330 DATA 25,145,32,145,211,88,96,0,24687
350 DATA 1,2,4,8,16,32,64,128,24942
360 DATA 160,144,255,170,85,85,144,3,25984
370 READ INIT:M1=INIT
380 FOR M3=0 TO 23: FOR M2=0 TO 7
390 READ M4: POKE INIT+M3*8+M2,M4
400 M1=M1+M4:NEXT
410 READ M4: IF M1<>M4 THEN PRINT"DATAFEHLER
IN ZEILE";M3*10+100:END
420 NEXT
430 READ E1,E2,E3,E4,E5,E6,E7,E8
440 M1=M1+E1+E2+E3+E4+E5+E6+E7+E8
450 READ M4: IF M1<>M4 THEN PRINT"DATAFEHLER
IN ZEILE 350":END
460 READ EX,EY,AUS,LINKS,RECHTS,EIN,M2,M3
470 M1=M1+EX+EY+AUS+LINKS+RECHTS+EIN+M2+M3
480 READ M4: IF M1<>M4 THEN PRINT"DATAFEHLER
IN ZEILE 360":END
485 POKE 1,M2 : POKE 2,M3
490 M1=INIT+4:M2=M1+4:M3=M2+4:M4=M3+4
500 SYS INIT

```

```

10 REM INTERFACE PROGRAMM FUER CBM 4XXX/8XXX
20 REM COPYRIGHT (C) ARTUR FISCHER FORSCHUNG 1984
30 REM AUFRUF DES PROGRAMMS MIT
40 REM SYS M1,EIN      SYS M1,AUS
50 REM SYS M1,LINKS    SYS M1,RECHTS
60 REM USR(E1)  USR(EX)  USR(EY)
70 REM M1 BIS M4 SIND MOTORANSTEUERUNGEN
80 REM E1 BIS E8 SIND TASTEREINGAENGE
90 REM EX UND EY SIND POTENTIOMETEREINGAENGE
100 DATA 834,0,169,0,240,41,169,3,208,1464
110 DATA 10,169,12,208,6,169,48,208,2294
120 DATA 2,169,192,120,133,187,32,245,3374
130 DATA 190,173,58,3,5,187,141,58,4189
140 DATA 3,32,212,200,138,37,187,133,5131
150 DATA 187,173,58,3,69,187,141,58,6007
160 DATA 3,168,169,63,141,67,232,162,7012
170 DATA 8,169,48,14,58,3,144,2,7458
180 DATA 9,4,141,65,232,9,8,141,8067
190 DATA 65,232,202,208,236,169,49,141,9369
200 DATA 65,232,140,58,3,88,96,12,30,10171
210 DATA 32,215,200,224,160,240,51,224,11517
220 DATA 144,240,47,134,187,169,50,141,12629
230 DATA 65,232,9,8,141,65,232,162,13543
240 DATA 8,10,44,65,232,16,2,9,13929
250 DATA 1,160,48,140,65,232,160,56,14791
260 DATA 140,65,232,202,208,235,37,187,16097
270 DATA 168,240,2,160,1,32,203,196,17099
280 DATA 88,96,160,80,142,65,232,162,18044
290 DATA 56,142,65,232,44,65,232,80,18960
300 DATA 4,200,208,248,136,76,7,3,19842
350 DATA 1,2,4,8,16,32,64,128,20097
360 DATA 160,144,255,170,85,85,209,2,21207
370 READ M1:INIT=M1
380 FOR M3=0 TO 20: FOR M2=0 TO 7
390 READ M4:POKE INIT+M3*8+M2,M4
400 M1=M1+M4:NEXT
410 READ M4: IF M1<>M4 THEN PRINT"DATAFEHLER
IN ZEILE";M3*10+100:PRINTM1:END
420 NEXT
430 READ E1,E2,E3,E4,E5,E6,E7,E8
440 M1=M1+E1+E2+E3+E4+E5+E6+E7+E8
450 READ M4: IF M1<>M4 THEN PRINT "DATAFEHLER
IN ZEILE 350":PRINT M1:END
460 READ EX,EY,AUS,LINKS,RECHTS,EIN,M2,M3
470 M1=M1+EX+EY+AUS+LINKS+RECHTS+EIN+M2+M3
480 READ M4: IF M1<>M4 THENPRINT"DATAFEHLER
IN ZEILE 360" : END
485 POKE1,M2 :POKE 2,M3
490 INIT=INIT+1:M1=INIT+4:M2=M1+4:M3=M2+4:M4=M3+4
500 SYS INIT

```

Programm keinen Kassettenbetrieb vornehmen. Es besteht jedoch keine Einschränkung beim Laden und Abspeichern von Programmen mit Kassette, da beim Programmstart das Maschinenprogramm aus den DATA-Zeilen neu erzeugt wird.

Bei dem C64 liegt das Programm in dem Freispeicherbereich CE00 bis CFFF.

Neben dem Maschinenprogramm selbst werden auch noch die genormten Parameter INIT M1, M2, M3, M4, RECHTS, LINKS, EIN, AUS, E1, E2, E3, E4, E5, E6, E7, E8, EX und EY gesetzt. Außerdem wird die Einsprungadresse der USR-Funktion festgelegt.

Für selbstgeschriebene BASIC-Programme sind daher unbedingt folgende Einschränkungen zu beachten:

Ähnlich wie Sie auch nicht die reservierten BASIC-Schlüsselworte wie PRINT oder STOP als Variablen verwenden dürfen, ist auch der Gebrauch der obigen Parameter als Variable verboten. Dabei beachten Sie bitte, daß Commodore BASIC Variablen desselben Typs nur anhand der ersten beiden Buchstaben unterscheidet. Namen wie

EXTRA, M114, EINMAL, LIFT, REIN...

sind ebenfalls verboten. Die Einschränkung betrifft jedoch nicht Variablen anderen Typs (Strings, Integer), so daß

EINGABES, M1%, AUTO\$...

durchaus vorkommen dürfen.

- Die USR-Funktion kann natürlich nicht mehr für andere Zwecke verwendet werden.
- Die obengenannten Speicherbereiche können nicht mehr für andere Zwecke verwendet werden.

Die letzte Funktion des Grundprogramms besteht in dem Einschalten des Interface und dem Ausschalten aller angeschlossenen Verbraucher. Dies wird durch den Befehl

500 SYS INIT

bewirkt. Damit wird das Interface sozusagen in betriebsbereitem Zustand an ein hier anschließendes Benutzerprogramm übergeben. Auch alle Beispielprogramme der Diskette sind nach diesem Muster aufgebaut. Studieren Sie diese, wenn Sie sich Anregungen holen wollen.

Bei der Benutzung der fischertechnik computing Software beachten Sie bitte noch folgendes:

Nicht alle Programme der fischertechnik computing Software passen in den Grundausbau des VC20 hinein. Gegebenenfalls ist hier eine Speichererweiterung vorzusehen. Dabei wird der 8k-Baustein in aller Regel genügen.

Das Programm für das Grafiktablett erfordert eine Ausstattung des Computers mit einem Grafiksystem. In dem C64 ist eine hochauflösende Grafik eingebaut; als Softwaresystem setzen wir SIMONS BASIC voraus. Besitzer anderer Computertypen und anderer Grafiksysteme können das C64 Programm GRAFIKTABLETT .64 als Richtschnur verwenden.

Das Diagnoseprogramm

Wenn Sie ein fischertechnik computing Modell aufgebaut haben, werden Sie vielleicht die Erfahrung machen, daß es nicht so läuft, wie Sie sich das vorgestellt haben. Wen wundert das bei dieser großen Zahl von Leitungen, die zwischen Modell und Interface hin- und herlaufen. Und wenn nur ein Taster vertauscht wäre, die verblüffendsten Effekte könnte dies zur Folge haben. Doppelt schwierig wird die Situation, wenn die Programme selbst geschrieben sind. Wo soll man da mit der Suche anfangen? In der Hardware oder der Software?

Damit Sie die Hardware eindeutig und komfortabel testen können, wurde das Diagnoseprogramm entwickelt. Es liegt auf der fischertechnik computing Diskette als DIAGNOSE... vor. Laden Sie dieses Programm immer zum Austesten eines Modells. Sie können mit ihm sämtliche Eingänge beobachten und feststellen, ob ihr Verhalten mit Ihren Vorstellungen übereinstimmt.

Mit den Zahlentasten suchen Sie einen Steuerausgang aus. Er wird auf dem Bildschirm invers angezeigt. Diesen angewählten Ausgang können Sie nun einschalten (Rechts- und Linkslauf) und ausschalten. Damit stellen Sie also nicht nur fest, ob ein Motor überhaupt läuft, sondern auch, ob er in der gewünschten Drehrichtung anläuft. Sollte dies nicht der Fall sein, so vertauschen Sie bitte die beiden Motoranschlüsse.

Mit C können Sie alle Motoren abschalten und mit X das Programm verlassen.

```

500 SYS INIT
600 REM
610 REM FISCHERTECHNIK COMPUTING
620 REM
630 REM DIAGNOSEPROGRAMM (C64 UND CBM 4XXX/8XXX)
640 REM
650 REM COPYRIGHT (C) ARTUR FISCHER FORSCHUNG 1984
660 REM
730 REM
800 REM FUNKTION
810 REM MIT DEM PROGRAMM WERDEN SAEMTLICHE
815 REM FUNKTIONEN DER MODELLE UEBERPRUEFT.
820 REM ALLE EINGAENGE WERDEN ANGEZEIGT.
830 REM ALLE AUSGAENGE WERDEN UEBER DIE
835 REM COMPUTERTASTATUR ANGESTEUERT.
900 PRINT CHR$(147)
910 PRINT"FISCHERTECHNIK"
920 PRINT"COMPUTING"
940 PRINT"DIAGNOSE"
950 PRINT
1000 DIM STA(4),STA$(4):REM STATUS DER 4 MOTOREN
1010 FOR I=1 TO 4
1020 LET STA(I)=AUS:STA$(I) = "AUS"
1030 NEXT I
1050 DIM M (4):REM MPGM.-ADRESSEN FUER MOTOR 1-4
1060 LET M(1)=M1:M(2)=M2:M(3)=M3:M(4)=M4
1070 LET M = 1:REM AKT.MOTOR
1110 PRINT"K O M M A N D O S "
1120 PRINT
1130 PRINT"C : ALLE MOTOREN AUS"
1140 PRINT"A : AKT.MOTOR AUS"
1150 PRINT"L : AKT.MOTOR LINKS"
1160 PRINT"R : AKT.MOTOR RECHTS"
1170 PRINT"X : PROGRAMMENDE"
1180 PRINT"1-4 : MOTORNr. WAEHLEN"
1190 PRINT:PRINT
1200 PRINT"  E1 E2 E3 E4 E5 E6 E7 E8  EX  EY"
1210 PRINT
1230 LET CLEAR$="          "
1240 PRINT "#####";
1250 PRINT TAB(2);CLEAR$:USR(E1);
1260 PRINT TAB(5);CLEAR$:USR(E2);
1270 PRINT TAB(8);CLEAR$:USR(E3);
1280 PRINT TAB(11);CLEAR$:USR(E4);
1290 PRINT TAB(14);CLEAR$:USR(E5);
1300 PRINT TAB(17);CLEAR$:USR(E6);
1310 PRINT TAB(20);CLEAR$:USR(E7);
1320 PRINT TAB(23);CLEAR$:USR(E8);
1330 PRINT TAB(26);CLEAR$:USR(EX);
1340 PRINT TAB(32);CLEAR$:USR(EY);
1400 GET K#
1410 IF K#=""THEN 2000

```

```

1500 IF K#="C"THEN FOR I=1 TO4:LET STA(I)=AUS
:STA$(I)="AUS":NEXT I:GOTO 2000
1510 IF K#="A"THEN LET STA(M)=AUS:STA$(M)="AUS"
:GOTO 2000
1520 IF K#="L"THEN LET STA(M)=LINKS:STA$(M)="LI."
:GOTO 2000
1530 IF K#="R"THEN LET STA(M)=RECHTS:STA$(M)="RE."
:GOTO 2000
1540 IF K#="X"THEN PRINT"␣":END
1550 M=VAL(K#)
1560 IF M<1 OR M>4 THEN LET M=1
2000 FOR I=1TO4
2010 SYS M(I),STA(I)
2015 IF I=M THEN PRINT CHR$(18);
2020 PRINT"#####";
2030 PRINT TAB(5*I-3);"M";I;"#####";
2040 PRINT"␣";
2050 PRINT STA(I);
2060 PRINT CHR$(146);
2070 NEXT I
2080 GOTO 1240

```

```

500 SYS INIT
600 REM
610 REM FISCHERTECHNIK COMPUTING
630 REM DIAGNOSEPROGRAMM (VC20)
650 REM COPYRIGHT (C) ARTUR FISCHER FORSCHUNG 1984
800 REM FUNKTION
820 REM ALLE EINGAENGE WERDEN ANGEZEIGT
830 REM ALLE AUSGAENGE WERDEN UEBER DIE
840 REM COMPUTERTASTATUR ANGESTEUERT.
900 PRINT CHR$(147)
910 PRINT"FISCHERTECHNIK"
920 PRINT"COMPUTING"
930 PRINT"DIAGNOSE"
950 PRINT
1000 DIM STA(4),STA$(4):REM STATUS DER 4 MOTOREN
1010 INPUT"X-POTI BENUTZT";X#
1020 X=LEFT$(X#,1)="J"
1030 INPUT"Y-POTI BENUTZT";Y#
1040 Y=LEFT$(Y#,1)="J"
1050 FOR I=1 TO 4
1060 LET STA(I)=AUS:STA$(I)="AUS"
1070 NEXT I
1080 DIM M(4):REM MPGM.-ADRESSEN FUER MOTOR 1-4
1090 LET M(1)=M1:M(2)=M2:M(3)=M3:M(4)=M4
1100 LET M=1:REM AKT.MOTOR
1105 PRINT CHR$(147)

```

```

1106 PRINT"FISCHERTECHNIK"
1107 PRINT"COMPUTING"
1108 PRINT"DIAGNOSE"
1110 PRINT"K O M M A N D O S"
1120 PRINT
1130 PRINT"C:ALLE MOTOREN AUS"
1140 PRINT"A:AKT.MOTOR AUS"
1150 PRINT"L:AKT.MOTOR LINKS"
1160 PRINT"R:AKT.MOTOR RECHTS"
1170 PRINT"X:PROGRAMMENDE"
1180 PRINT"1-4:MOTORNr. WAEHLEN"
1190 PRINT
1200 PRINT"  E1 E2 E3 E4  EX"
1210 PRINT:PRINT
1220 PRINT"  E5 E6 E7 E8  EY"
1230 PRINT
1240 LET CLEAR$="          "
1250 PRINT "#####";
1260 PRINT TAB(2);CLEAR$:USR(E1);
1270 PRINT TAB(5);CLEAR$:USR(E2);
1280 PRINT TAB(8);CLEAR$:USR(E3);
1290 PRINT TAB(11);CLEAR$:USR(E4);
1300 IF X THEN PRINT TAB(16);CLEAR$:USR(EX);
1310 PRINT:PRINT:PRINT
1320 PRINT TAB(2);CLEAR$:USR(E5);
1330 PRINT TAB(5);CLEAR$:USR(E6);
1340 PRINT TAB(8);CLEAR$:USR(E7);
1350 PRINT TAB(11);CLEAR$:USR(E8);
1360 IF Y THEN PRINT TAB(16);CLEAR$:USR(EY);
1400 GET K#
1410 IF K#=""THEN 2000
1500 IF K#="C"THEN FOR I=1 TO4:LET STA(I)=AUS
:STA$(I)="AUS":NEXT I:GOTO2000
1510 IF K#="A"THEN LET STA(M)=AUS:STA$(M)="AUS"
:GOTO 2000
1520 IF K#="L"THEN LET STA(M)=LINKS:STA$(M)="LI."
:GOTO2000
1530 IF K#="R"THEN LET STA(M)=RECHTS:STA$(M)="RE."
:GOTO2000
1540 IF K#="X"THEN PRINT"␣":END
1550 M=VAL(K#)
1560 IF M<1 OR M>4 THEN LET M=1
2000 FOR I=1TO4
2010 SYS M(I),STA(I)
2015 IF I=M THEN PRINT CHR$(18);
2020 PRINT"#####";
2030 PRINT TAB(5*I-3);"M";I;"#####";
2040 PRINT"␣";
2050 PRINT STA(I);
2060 PRINT CHR$(146);
2070 NEXT I
2080 GOTO 1250

```

Commodore BASIC

Die fischertechnik computing Programme der beigefügten Diskette sind in Commodore BASIC geschrieben. Dies trifft auch für die in den fischertechnik computing Programmieranleitungen dokumentierten Programme zu. Dabei wurde darauf geachtet, daß möglichst wenige computerspezifische Merkmale einfließen. Ganz konnte es jedoch nicht vermieden werden. In allen Zweifelsfällen gehen Sie davon aus, daß die Beispielprogramme auf einem C64 entwickelt wurden. Bei den anderen Computertypen ergeben sich eventuell Unterschiede in dem Bildschirmformat.

All diese Änderungen sind jedoch in den Programmen auf Diskette bereits durchgeführt. Auch können die Diskettenprogramme noch in andere Details von der gedruckten Dokumentation abweichen, wo sich Vorteile durch die Ausnutzung computerspezifischer Merkmale ergaben.

Checkliste

Sollte das fischertechnik computing Interface sich einmal widerwillig zeigen und nicht so arbeiten, wie Sie es erwarten, so überprüfen Sie bitte folgende Punkte mit dem Diagnoseprogramm:

Das Interface zeigt bei E1 bis E8 überall 1 an, obwohl kein Modell angeschlossen ist. – Das Interface ist nicht an den Computer oder nicht an das Netzgerät angeschlossen.

Einer der Eingänge E1 bis E8 zeigt bei Betätigung des Tasters gerade das umgekehrte Ergebnis. – Öffner- und Schließfunktion des Tasters sind vertauscht.

Einer der Eingänge E1 bis E8 zeigt immer das Ergebnis 0, obwohl er angeschlossen ist und betätigt wird. – Prüfen Sie die Verkabelung.

Einer der Eingänge E1 bis E8 zeigt immer das Ergebnis 1, selbst wenn kein Modell angeschlossen ist. – Vermutlich das Eingangsgatter IC 4014 durch Überspannung (elektrostatische Aufladung) beschädigt. Ein Motorausgang arbeitet nicht. – Bitte Verkabelung überprüfen.

Ein Motorausgang geht nur in einer Richtung. – Leistungsstufe des Motors defekt.

Motor läuft sehr langsam oder setzt aus. – Entweder Netzgerät durch zu viele Motoren überlastet (zwei Netzgeräte verwenden) oder Netzgerät bei Verwendung des regelbaren Ausgangs nicht voll aufgedreht.

C64 und VC20: Potentiometerwert ist größer als 255. – Bei Verwendung anderer Geber als den Potentiometern (mit höherem Widerstandswert) kann eine größere Anzeige erscheinen. Dies ist kein Defekt. Allerdings dauert die Analogmessung eine merkbare Zeit.

VC20: Der Computer „hängt“, die Stop-Taste ist wirkungslos. – Bitte Verkabelung prüfen, es wurde ein Analogwert verlangt, obwohl kein Geber angeschlossen war.

Bei Defekten schicken Sie das Gerät bitte an die Fischerwerke, Abt. Service, ein.

Technische Daten

fischertechnik computing Interface Commodore C64/VC20, Art.-Nr. 30562, und Commodore 4xxx/8xxx, Art.-Nr. 30561.

4 Ausgänge zum Anschluß von Motoren, Lampen, Elektromagneten.. (M1 bis M4).

Polarität des Ausgangs steuerbar.

Belastbarkeit: 1A Dauerstrom, 1,5A Spitzenstrom.

8 Eingänge für digitale Signale (E1 bis E8).

Durch interne Beschaltung sowohl Anschluß von elektromagnetischen Artikeln (Taster, Schalter, Relais) in positiver Logik als auch Anschluß von TTL-Ausgängen möglich. Schutz gegen Überspannung eingebaut.

2 Eingänge für analoge Signale (EX und EY).

Anschließbar sind Geber mit Widerstandswerten zwischen 0 und 5 k Ω , z. B. Potentiometer, Fotowiderstände...

Überwachungsschaltung des Datenstroms. Bei Ausbleiben von Datensignalen des Commodore Computers schaltet das Interface nach 0,5 Sekunden alle Ausgänge inaktiv. Die Signale bleiben jedoch gespeichert.

Überwachungsschaltung der Software. Bei gravierenden Syntaxfehlern spricht ebenfalls die Überwachungsschaltung, jedoch ohne Verzögerung, an. Die Überwachungsschaltung reagiert auch auf Unterversorgung des Interface, sei es durch Überlastung oder zu niedrige Spannung des Netzgeräts. Software mit Interfacesteuerungsbefehlen und Beispielprogrammen für die fischertechnik computing Modelle im Lieferumfang enthalten.


Benutzung von fischertechnik Elektromechanik und Elektronik

Das fischertechnik computing Interface ist kompatibel zu den Bauteilen und Elektronikbausteinen der obengenannten Baukästen. Anstelle der bei den fischertechnik computing Modellen verwendeten mini-Taster können Sie genauso gut Taster und Schalter anderer Bauart anschließen. Z.B. den großen Taster oder den Polwendeschalter, aber auch den Reedkontakt oder den Schaltkontakt eines Relais. Aufpassen müssen Sie jedoch bei der Verwendung von selbstgebauten Tastern und Schaltern aus Gelenkbausteinen und Federn. Hier könnten eventuell Prellerscheinungen auftreten. Wir empfehlen, in diesen Fällen den Taster mehrmals abzufragen und den Wert nur dann als gültig zu erachten, wenn zweimal hintereinander der gleiche Wert erschien.

Die Analogeingänge des Interface können mit Sensoren beschaltet werden, die einen Widerstands-wert zwischen 0 und 5 k Ω als Ausgang liefern. Zunächst bieten sich die Potentiometer aus dem Baukasten fischertechnik computing an. Genauso können aber auch andere Bauelemente, wie z. B. der Fotowiderstand, verwendet werden.

Die Motorausgänge des Interface sind kräftig belastbar. Nicht nur die mini-Motoren, auch der S-Motor und der N-Motor lassen sich mit dem Interface ansteuern, wobei noch eine Lampe zur Funktions-anzeige parallelgeschaltet sein darf. Außer Motoren eignen sich noch der Elektromagnet und das Relais RBII.

Die Signale der Elektronikbausteine mit integriertem Schaltkreis aus der TTL-Familie (z.B. Schwell-wertschalter) können ebenfalls in die Eingänge des Interface eingespeist werden. Als gemeinsamer Bezugspunkt ist jedoch zuvor die Masse-schiene des Elektronikbausteins mit der Massebuche-se des Interface zu verbinden. In der Abb. 2 zeigen wir, wie eine Lichtschranke aufgebaut wird. Der Schwell-wertschalter dient dazu, die Ansprechschwelle der Lichtschranke einzustellen.



Verdrahtungsplan der Interface Ein- und Ausgänge

- E1** braun · brown · brun
- E2** rot · red · rouge
- EX** orange · orange · orange
- EY** gelb · yellow · jaune
- +5V** grün · green · vert
- E3** blau · blue · bleu
- E4** violett · violet · violet
- E5** grau · grey · gris
- E6** weiß · white · blanc
- E7** schwarz · black · noir
- E8** braun · brown · brun
- +5V** rot · red · rouge
- M1** orange · orange · orange
- M1** gelb · yellow · jaune
- M2** grün · green · vert
- M2** blau · blue · bleu
- M3** violett · violet · violet
- M3** grau · grey · gris
- M4** weiß · white · blanc
- M4** schwarz · black · noir

**Download von www.ft-fanarchiv.de
gescannt von Peter Remm**

